# Maintainer: Balló György <ballogyor+arch at gmail dot com>

pkgname=gnome-break-timer
pkgver=2.1.0
pkgrel=1
pkgdesc="Keeps track of how much you are using the computer, and it reminds you to take regular breaks"
arch=('loong64' 'x86_64')
url="https://wiki.gnome.org/Apps/BreakTimer"
license=('GPL3')
depends=('gsound' 'gtk3' 'libnotify')
makedepends=('meson' 'git' 'vala')
_commit=8f938244d5f444bde2c47c510d9245f0e6d768cf
source=("git+https://gitlab.gnome.org/GNOME/gnome-break-timer.git#commit=$_commit")
sha256sums=('SKIP')

pkgver() {
  cd $pkgname
  git describe --tags | sed 's/-/+/g'
}

build() {
  arch-meson $pkgname build
  meson compile -C build
}

package() {
  DESTDIR="$pkgdir" meson install -C build
}
